# $Id: PKGBUILD 184454 2013-05-06 19:40:24Z foutrelis $
# Maintainer: Eric Bélanger <eric@archlinux.org>

pkgname=xine-ui
pkgver=0.99.7
pkgrel=2
pkgdesc="A free video player for Unix"
arch=('i686' 'x86_64')
license=('GPL')
url="http://www.xine-project.org"
depends=('xine-lib' 'curl' 'libpng' 'libxft' 'xdg-utils' 'shared-mime-info' 
         'hicolor-icon-theme' 'desktop-file-utils' 'lirc-utils')
options=('!strip')
install=xine-ui.install
source=(http://downloads.sourceforge.net/xine/${pkgname}-${pkgver}.tar.xz)
sha1sums=('ca82e09cf8245d038813bc4dbd8384ba79cbf197')

build() {
  cd ${pkgname}-${pkgver}
  ./configure --prefix=/usr --mandir=/usr/share/man \
    --with-x --enable-lirc --without-aalib --enable-debug
  make
}

package() {
  cd ${pkgname}-${pkgver}
  make DESTDIR="${pkgdir}" desktopdir=/usr/share/applications install
}
